# 11　方程式

*Mathematica* は方程式を厳密に，あるいは数値的に解くことができる．そのために， Solve, NSolve, FindRoot などが用意されている．また微分方程式の厳密解あるいは数値解を求めるには DSolve, NDSolve を用いる．

## ■ 変換規則

この章の説明をするために，変換規則という *Mathematica* 独自の用語について説明する．

## ■ 変換規則（ルール）

x->a は， x を a に変換するという規則を表す，変換規則（ルール）である．

変換規則は単なる規則であるから，これだけを実行してもなにも影響はでない．特に x = a という割り当てとは違って，変数 x の値は何も変化しない．

```
In[1]:= x -> a
```

```
Out[1]= x → a
```

```
In[2]:= x
```

```
Out[2]= x
```

## ■ 変換規則の適用

expr /. x->a は x を a に変換するという変換規則を expr に適用して得られた結果を返す．

"/." は，スラッシュ "/" とピリオド "." である．その間に空白を入れてはいけない．

```
In[3]:= x /. x -> a
```

```
Out[3]= a
```

```
In[4]:= x^2 + Sin[x] /. x -> a
```

```
Out[4]= a^2 + Sin[a]
```

この計算を行った後も，変数 x の値や式 x^2+Sin[x] には変化はない．

```
In[5]:= x
```

```
Out[5]= x
```

```
In[6]:= x^2 + Sin[x]
```

```
Out[6]= x^2 + Sin[x]
```

## ■ 方程式

方程式は等号で左右の辺を結び付けたものであるが， *Mathematica* では，この等号を == で表す．これは変数への割り当て = と区別するためである．

```
In[7]:= 2 x^2 - x == 2
```

$$-x + 2\,x^2 == 2$$

連立方程式は，方程式のリストで表される．

```
In[8]:= {3 x + 2 y == 1, x - 3 y == 5}
```

$$\{3\,x + 2\,y == 1,\ x - 3\,y == 5\}$$

方程式が変数を含まず，真偽がはっきりしている場合には，その真偽値が返る．

```
In[9]:= {2 + 4 == 6, 3 + 5 == 7}
```

```
Out[9]= {True, False}
```

方程式の変数に，ルールを用いて具体的な値を代入すれば，具体的な真偽値が返る．

```
In[10]:= {3 x + 2 y == 1, x - 3 y == 5} /. {x -> 2, y -> -1}
```

```
Out[10]= {False, True}
```

## ■ 方程式のすべての解を求める

方程式のすべての解を理論的に求めるには Solve または NSolve を用いる

## ■ 厳密解を求める

Solve は方程式のすべての解を，理論的に，厳密に求めようとする．

Solve[eqn, x] は方程式 eqn を x について理論的に解いた厳密解を，変換規則を用いて返す．

Solve[{eqn1, eqn2, ...}, {x1, x2, ...}] は連立方程式 eqn1, eqn2, ... を x1, x2, ... について理論的に解いた厳密解を，変換規則を用いて返す．

方程式 $2\,x^2 - x - 2 = 0$ を解いてみる．

```
In[11]:= Solve[2 x^2 - x - 2 == 0, x]
```

$$\left\{\left\{x \to \frac{1}{4}\left(1 - \sqrt{17}\right)\right\}, \left\{x \to \frac{1}{4}\left(1 + \sqrt{17}\right)\right\}\right\}$$

解は 2 個ある．解がリストのリストで表されているのには理由があるが，後で説明する．

この計算結果が合っていることを， 2 x^2 - x - 2 にこの変換規則を適用して確かめてみる．

```
In[12]:= 2 x^2 - x - 2 /. %
```

$$\left\{-2 + \frac{1}{8}\left(1 - \sqrt{17}\right)^2 + \frac{1}{4}\left(-1 + \sqrt{17}\right), -2 + \frac{1}{4}\left(-1 - \sqrt{17}\right) + \frac{1}{8}\left(1 + \sqrt{17}\right)^2\right\}$$

これが，$x \to \frac{1}{4}\left(1-\sqrt{17}\right)$ という変換規則と $x \to \frac{1}{4}\left(1+\sqrt{17}\right)$ という変換規則を `2 x^2 - x - 2` に施した結果である．これが実際には `0` であることは，`Simplify` を用いて簡約化することで分る．

```
In[13]:= Simplify[%]

Out[13]= {0, 0}
```

連立方程式 $2x + y^2 - xy = 10$, $x^2 - y = 5$ の解は次のようにして求められる．

```
In[14]:= Solve[{2 x + y^2 - x y == 10, x^2 - y == 5}, {x, y}]

Out[14]= {{x → -1, y → -4}, {x → 3, y → 4},
           {x → 1/2 (-1 - Sqrt[21]), y → 1/2 (1 + Sqrt[21])}, {x → 1/2 (-1 + Sqrt[21]), y → 1/2 (1 - Sqrt[21])}}
```

この方程式の解は 4 組あることがわかる．このように，`Solve` が返す方程式の解は，解の組のリストで与えられる．得られた解を左辺に代入して確かめてみる．

```
In[15]:= Simplify[{2 x + y^2 - x y, x^2 - y} /. %]

Out[15]= {{10, 5}, {10, 5}, {10, 5}, {10, 5}}
```

`Solve` は多項式からなる方程式だけではなく，対数関数や三角関数を含んだ方程式の厳密解を求めることができる．次は $\log(x^2+x+1)=2$ の解である．

```
In[16]:= Solve[Log[x^2 + x + 1] == 2, x]

Out[16]= {{x → 1/2 (-1 - Sqrt[-3 + 4 E^2])}, {x → 1/2 (-1 + Sqrt[-3 + 4 E^2])}}
```

しかし，理論的に厳密解を求めることが不可能な場合もある．

```
In[17]:= Solve[Log[x^2 + x + 1] == x, x]

        Solve::tdep : 本質的に非代数的な方法で解かれる変数が方程式に含まれているようです. 詳細

Out[17]= Solve[Log[1 + x + x^2] == x, x]
```

多項式の方程式は，`4` 次方程式までは理論的に解くことができる．

```
In[18]:= Solve[x^4 - 3 x^3 + x - 5 == 0, x]
```

Out[18]=

$$\left\{\left\{x \to \frac{3}{4} - \frac{1}{2}\sqrt{\frac{9}{4} + \frac{\left(-198+\sqrt{53943}\right)^{1/3}}{3^{2/3}} - \frac{17}{\left(3\left(-198+\sqrt{53943}\right)\right)^{1/3}}} - \frac{1}{4}\sqrt{\left|18 - \frac{4\left(-198+\sqrt{53943}\right)^{1/3}}{3^{2/3}} + \frac{68}{\left(3\left(-198+\sqrt{53943}\right)\right)^{1/3}} - \frac{19}{\sqrt{\frac{9}{4} + \frac{\left(-198+\sqrt{53943}\right)^{1/3}}{3^{2/3}} - \frac{17}{\left(3\left(-198+\sqrt{53943}\right)\right)^{1/3}}}}\right|}\right\},\right.$$

$$\left\{x \to \frac{3}{4} - \frac{1}{2}\sqrt{\frac{9}{4} + \frac{\left(-198+\sqrt{53943}\right)^{1/3}}{3^{2/3}} - \frac{17}{\left(3\left(-198+\sqrt{53943}\right)\right)^{1/3}}} + \frac{1}{4}\sqrt{\left|18 - \frac{4\left(-198+\sqrt{53943}\right)^{1/3}}{3^{2/3}} + \frac{68}{\left(3\left(-198+\sqrt{53943}\right)\right)^{1/3}} - \frac{19}{\sqrt{\frac{9}{4} + \frac{\left(-198+\sqrt{53943}\right)^{1/3}}{3^{2/3}} - \frac{17}{\left(3\left(-198+\sqrt{53943}\right)\right)^{1/3}}}}\right|}\right\},$$

$$\left\{x \to \frac{3}{4} + \frac{1}{2}\sqrt{\frac{9}{4} + \frac{\left(-198+\sqrt{53943}\right)^{1/3}}{3^{2/3}} - \frac{17}{\left(3\left(-198+\sqrt{53943}\right)\right)^{1/3}}} - \frac{1}{4}\sqrt{\left|18 - \frac{4\left(-198+\sqrt{53943}\right)^{1/3}}{3^{2/3}} + \frac{68}{\left(3\left(-198+\sqrt{53943}\right)\right)^{1/3}} + \frac{19}{\sqrt{\frac{9}{4} + \frac{\left(-198+\sqrt{53943}\right)^{1/3}}{3^{2/3}} - \frac{17}{\left(3\left(-198+\sqrt{53943}\right)\right)^{1/3}}}}\right|}\right\},$$

$$\left.\left\{x \to \frac{3}{4} + \frac{1}{2}\sqrt{\frac{9}{4} + \frac{\left(-198+\sqrt{53943}\right)^{1/3}}{3^{2/3}} - \frac{17}{\left(3\left(-198+\sqrt{53943}\right)\right)^{1/3}}} + \frac{1}{4}\sqrt{\left|18 - \frac{4\left(-198+\sqrt{53943}\right)^{1/3}}{3^{2/3}} + \frac{68}{\left(3\left(-198+\sqrt{53943}\right)\right)^{1/3}} + \frac{19}{\sqrt{\frac{9}{4} + \frac{\left(-198+\sqrt{53943}\right)^{1/3}}{3^{2/3}} - \frac{17}{\left(3\left(-198+\sqrt{53943}\right)\right)^{1/3}}}}\right|}\right\}\right\}$$

この計算結果はとても長い表記になるので，省略する．

しかし， 5 次方程式以上になると，その解を四則演算と根号とで表すことができなくなる．このような場合 Solve はその方程式の解を Root という関数を用いて表した結果を返す．

```
In[19]:= Solve[x^5 - 3 x^3 + x - 5 == 0, x]

Out[19]= {{x → Root[-5 + #1 - 3 #1^3 + #1^5 &, 1]},
          {x → Root[-5 + #1 - 3 #1^3 + #1^5 &, 2]}, {x → Root[-5 + #1 - 3 #1^3 + #1^5 &, 3]},
          {x → Root[-5 + #1 - 3 #1^3 + #1^5 &, 4]}, {x → Root[-5 + #1 - 3 #1^3 + #1^5 &, 5]}}
```

$-5 + \#1 - 3\,\#1^3 + \#1^5$ & は #1 を引き数とする純関数であり， #1 に x を入れると方程式の左辺となる． Root[f, n] は f == 0 という方程式の n 番目の解を表す．この Solve の結果は何も言っていないようであるが，解が 5 個ありそれがすべてである，ということが重要なのである．

> 練習問題： $x^2 - 3x - 2 = 0$ の解を求めよ．

## ■ 数値解を求める

NSolve はすべての解の近似値（数値解）を求めようとする．

NSolve[eqn, x] は方程式 eqn のすべての数値解を，変換規則を用いて返す．

```
In[20]:= NSolve[x^4 - 3 x^3 + x - 5 == 0, x]

Out[20]= {{x → -1.14039}, {x → 0.536692 - 1.06842 i}, {x → 0.536692 + 1.06842 i}, {x → 3.067}}
```

求められた解を x^4 - 3 x ^3 + x - 5 に適用してみる．

```
In[21]:= x^4 - 3 x^3 + x - 5 /. %

Out[21]= {1.77636 × 10^-15, 1.44329 × 10^-15 - 4.44089 × 10^-16 i, 1.44329 × 10^-15 + 4.44089 × 10^-16 i, 0.}
```

すべて非常に小さい値となっている．このような 0 に非常に近い値を 0 に切り捨てる関数 Chop を用いてみる．

```
In[22]:= Chop[%]

Out[22]= {0, 0, 0, 0}
```

$\log(x^2 + x + 1) - 2 = 0$ という方程式を解いてみる．

```
In[23]:= NSolve[Log[x^2 + x + 1] - 2 == 0, x]

Out[23]= {{x → -3.07664}, {x → 2.07664}}
```

左辺のグラフを見てみると，確かに -3.07664 と 2.07664 のあたりに解があることがわかる．

```
In[24]:= Plot[Log[x^2 + x + 1] - 2, {x, -4, 3}]
```




```
Out[24]= - Graphics -
```

では $\log x = \cos(3x)$ という方程式はどうか．関数のグラフを見ると，解が 3 個あることがわかる．

```
In[25]:= Plot[{Log[x], Cos[3 x]}, {x, 0, 5}]
```




```
Out[25]= - Graphics -
```

しかし， NSolve はその解を見つけることができない．

```
In[26]:= NSolve[Log[x] == Cos[3 x], x]

Solve::tdep : 本質的に非代数的な方法で解かれる変数が方程式に含まれているようです. 詳細

Out[26]= NSolve[Log[x] == Cos[3 x], x]
```

これは， NSolve は代数的な手法を用いて完全な解の集合を求めようとするからである．

> 練習問題：方程式 $x^5 - x^3 + 3x^2 - 2 = 0$ のすべての解の近似値を求めよ．

## ■ 方程式の解を反復法を用いて求める

すべての解を求めるのではなくて，とりあえず解が一つだけ欲しいときがある．そのようなときには，FindRoot を用いると手早く数値解が見つけられる．

FindRoot[eqn, {x, x0}] は， x=x0 を初期値として反復法を用いて方程式 eqn の数値解を一つだけ返す．

先ほどの方程式 $\log x = \cos(3x)$ の解を初期値 x0 を 1 にして求めると次の解が求められる．

```
In[27]:= FindRoot[Log[x] == Cos[3 x], {x, 1}]

Out[27]= {x → 0.664145}
```

他の解を求めるには，色々な初期値で計算する必要がある．

```
In[28]:= FindRoot[Log[x] == Cos[3 x], {x, 2}]

Out[28]= {x → 1.77436}
```

```
In[29]:= FindRoot[Log[x] == Cos[3 x], {x, 2.5}]

Out[29]= {x → 2.29197}
```

このように，FindRoot を用いてすべての解を求めるには，グラフを描いておいて解の見当をつけてから始める必要がある．

FindRoot[{eqn1, eqn2, ...}, {x, x0}, {y, y0}, ...] は連立方程式 eqn1, eqn2, ... の一組の数値解を x=x0, y=y0, ... を初期値とした反復法で求める．

連立方程式 $\sin(x) - \cos(y) + x + y - 1 = 0$, $\cos(x) + \sin(x + y) - x^2 - 1 = 0$ を解いてみる．まず，解の見当をつけるために，$x$, $y$ 平面上で各々の等式の解となっている曲線を表示してみる．それには左辺の値が 0 である等高線を引けばよい．

```
In[30]:= ContourPlot[Sin[x] - Cos[y] + x + y - 1, {x, -4, 4},
          {y, -4, 4}, Contours -> {0}, ContourShading -> False]
```




```
Out[30]= - ContourGraphics -
```

```
In[31]:= ContourPlot[ Cos[x] + Sin[x + y] - x^2 - 1, {x, -4, 4},
          {y, -4, 4}, Contours -> {0}, ContourShading -> False]
```




```
Out[31]= - ContourGraphics -
```

これらを重ねて表示したときの交点が，この連立方程式の解である．

```
In[32]:= Show[%, %%]
```




```
Out[32]= - Graphics -
```

これを見ると解が 2 組あることがわかる．これらを FindRoot で求めるには，目標の解に近い初期値から始めればよい．

```
In[33]:= FindRoot[
          {Sin[x] - Cos[y] + x + y - 1 == 0, Cos[x] + Sin[x + y] - x^2 - 1 == 0}, {x, -1}, {y, 2}]

Out[33]= {x → -0.802863, y → 2.05591}

In[34]:= FindRoot[
          {Sin[x] - Cos[y] + x + y - 1 == 0, Cos[x] + Sin[x + y] - x^2 - 1 == 0}, {x, 1}, {y, 1}]

Out[34]= {x → 0.795694, y → 0.40791}
```

## ■ 微分方程式

未知関数 y[x] の導関数は y'[x] で，2 階の導関数は y''[x] で表す．微分方程式は y''[x]+y'[x]==2y[x] のように表される．微分方程式を厳密に解くには DSolve を用いる．また，数値的に解くには NDSolve を用いる．

## ■ 微分方程式を厳密に解く

DSolve[eqn, y[x], x] は， x の関数 y[x] についての微分方程式 eqn の厳密解を返す．解に含まれる積分定数は C[i] で表される．

微分方程式 $y'' + y' - 2y = 0$ を解いてみる． DSolve に与える微分方程式は， y の式ではなく y[x] という形で変数を明示する必要がある．

```
In[35]:= DSolve[y''[x] + y'[x] - 2 y[x] == 0, y[x], x]

Out[35]= {{y[x] → ℯ^(-2 x) C[1] + ℯ^x C[2]}}
```

この結果は，微分方程式の解は $y = C_1\, e^{-2x} + C_2\, e^{x}$ （ただし $C_1$, $C_2$ は任意定数）であることを意味している．

DSolve[{eqn1, eqn2, ...}, {y1[x], y2[x], ...}, x] は連立微分方程式の厳密解を返す．

連立微分方程式 $y_1' + y_2 = 0$, $y_1 - y_2' = 0$ の解は次のとおり $y_1 = C_1 \cos x - C_2 \sin x$, $y_2 = C_2 \cos x + C_1 \sin x$ である．

```
In[36]:= DSolve[{y1'[x] + y2[x] == 0, y1[x] - y2'[x] == 0}, {y1[x], y2[x]}, x]

Out[36]= {{y1[x] → C[1] Cos[x] - C[2] Sin[x], y2[x] → C[2] Cos[x] + C[1] Sin[x]}}
```

また微分方程式の初期条件は，連立方程式の形で指定できる．上の微分方程式に $y_1(0) = 1,\ y_2(0) = -1$ という初期条件をつけると，次のように積分定数が具体的な値となった解が返る．

```
In[37]:= DSolve[{y1'[x] + y2[x] == 0, y1[x] - y2'[x] == 0, y1[0] == 1, y2[0] == -1},
          {y1[x], y2[x]}, x]

Out[37]= {{y1[x] → Cos[x] + Sin[x], y2[x] → -Cos[x] + Sin[x]}}
```

DSolve はどのような微分方程式でも解けるわけではない．たとえば $y'' + y\sin(x) = \log x$ は解けない．そのような場合には入力そのままの式を返す．

```
In[38]:= DSolve[y''[x] + y[x] Sin[x] == Log[x], y[x], x]

Out[38]= DSolve[Sin[x] y[x] + y''[x] == Log[x], y[x], x]
```

注意： y''[x]==0 と入力すべきところを y''[x]=0 と入力した場合， y''[x] に 0 が割り当てられるので，それ以降は y''[x] を用いた計算が正常に行なえなくなる．このような場合には，y''[x]=. を実行して割り当てを解除してから計算を行う．

## ■ 微分方程式の数値解を求める NDSolve

微分方程式 $y'' + y\sqrt{x} + \sin(x) = 0$ のような厳密な解を求めることができないものも，初期条件を与えるて NDSolve を用いると数値解を求めることができる．

NDSolve[{eqn1, eqn2, ...}, y[x], {x, xmin, xmax}] は， x の関数 y[x] について，初期条件も含めた微分方程式 eqn1, eqn2, ... の数値解を求める．

微分方程式 $y'' + y\sqrt{x} + \sin(x) = 0$ と初期条件 $y(0) = 0$, $y'(0) = 0$ を満たす数値解を， $x$ の範囲が 0 から 10 の間で求める．

```
In[39]:= sol =
          NDSolve[{y''[x] + y[x] Sqrt[x] + Sin[x] == 0, y[0] == 0, y'[0] == 0}, y[x], {x, 0, 10}]

Out[39]= {{y[x] → InterpolatingFunction[{{0., 10.}}, <>][x]}}
```

答えはリストのリストになっているので，この解から y[x] を引き出すには， y[x] /. sol[[1]] とすればよい．求められたものは補間によって値が決まる関数（interpolating function）に過ぎない． Plot を用いれば，解のグラフが見える．このとき，関数に Evalutate を施しておいた方が，描画速度が速くなる．

```
In[40]:= Plot[Evaluate[y[x] /. sol[[1]]], {x, 0, 10}]
```




```
Out[40]= - Graphics -
```

この解のグラフと $y = \frac{x}{6}$ のグラフとの交点の $x$ 座標を FindRoot を用いて求めてみよう．

```
In[41]:= Plot[{Evaluate[y[x] /. sol[[1]]], x / 6}, {x, 0, 10}]
```




```
Out[41]= - Graphics -

In[42]:= FindRoot[Evaluate[y[x] /. sol[[1]]] == x / 6, {x, 4}]

Out[42]= {x → 4.30436}

In[43]:= FindRoot[Evaluate[y[x] /. sol[[1]]] == x / 6, {x, 6}]

Out[43]= {x → 5.93713}
```

## ■ 演習問題

[11-1] 方程式 $x^5 - x^4 - 3x^2 + 2 = 0$ の数値解を NSolve を用いて求て，この方程式には何個の実数解があるか答えよ．

[11-2] 方程式 $\sin(x) + \cos(x^2) = x$ のすべての実数解の近似値を求めよ．

[11-3] 連立方程式 $\sin(x) - y^2 = 0, \sin(x + y) - x^2 + y = 0$ の実数値解の近似値をすべて求めよ．

[11-4] 微分方程式 $y'' - 2y' + y + \sin(x) = 0$ の解を求めよ．

[11-5] 単振動の微分方程式は $y'' + y = 0$ である．これに $y(0) = 0, y'(0) = 1$ という初期条件を付けて DSolve を用いて解を求めよ．また，その解のグラフを Plot を用いて描け．

[11-6] 時計の振り子の運動は厳密には単振動ではなく， $y'' + \sin(y) = 0$ という微分方程式で表される．これに $y(0) = 0, y'(0) = 1$ という初期条件をつけたものの数値解を NDSolve を用いて求めて，それのグラフを描け．さらに，それを [11-5] で描いたグラフと共に同一画面上に描け．そして，時計の振り子の実際の動きは，それが単振動であると考えた場合に比べて，速いか遅いかを答えよ．